
1 Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.3.1.16 which is part of the ATS iwd-TTCN3-B2019-06_D20wk11 test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Fu Wang

wangfu@datangmobile.cn
1.1 Verification Test Summary

Test Case: 
8.1.3.1.16
ATS Version:
iwd-TTCN3-B2019-06_D20wk11
System Simulator used:
Datang LinkTester 5G Protocol Conformance Test System ECT9610
UE used:
HiSilicon Balong 5000
Verification Status:
PASS
2 Corrections required

2.1 Change 1

	Function name
	f_TC_8_1_3_1_16_TestBody()

	Reason for change
	1. For the configured report amount is infinity, the de-registration procedure may be interrupted by the measurement reports after step8, which causes the TC to fail. The IE report Interval can be longer to ensure that no measurement report is transmitted during the de-registration procedure.
2. Given that reportInterval was changed, the timer at step 4 should be changed (increased) together in order not to pass a non-conformant UE.
Note: it needs a prose CR to support.

	Summary of change
	1. Modify the default value of report Interval ms480 to ms10240
2. Increase timer in step 4 from 10s to 30s(More than 10.24s)

	TTCN module
	NR5GC\8_1_3\ RRC_Measurement_NR5GC.ttcn.ttcn

	MCC160 Comment
	


Before change

function f_TC_8_1_3_1_16_TestBody() runs on NR_BASE_PTC

  {

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NR_CellPowerLevels_Type v_CellPowerLevels;

    var NR_AbsoluteCellPower_Type v_CellPower_A3_NoEntry_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_A3_Entry_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_ServingCell_FR1;

    var RSRP_Range v_RsrpThres;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour2;

    var PhysCellId v_PhysCellIdNeighbour4;

    var template (value) MeasObjectNR v_MeasObjectNR;

    var template (omit) MeasObjectToAddModList v_MeasObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    v_CellPower_ServingCell_FR1 := -85;

    v_CellPower_A3_NoEntry_FR1 := -91;

    v_CellPower_A3_Entry_FR1 := -78;

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_ServingCell_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),

                                cs_NR_CellPower(nr_Cell2, v_CellPower_A3_NoEntry_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),

                                cs_NR_CellPower(nr_Cell4, v_CellPower_A3_NoEntry_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell2, v_CellPower_A3_Entry_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    v_CellPowerLevels := {v_CellPowerList_AtT0, v_CellPowerList_AtT1};

    f_NR_SetCellPowerList(v_CellPowerLevels[0]);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup NR measurement and reporting for intra-frequency event A3 (measId 1)

    //The UE transmits an RRCReconfigrationComplete message

    v_RsrpThres := 0;

    v_MeasObjectNR := f_NR_MeasObjectNR_SSB (nr_Cell1, omit, omit);

    v_MeasObjectNR.whiteCellsToAddModList := {cs_PCI_RangeElement(900)};  //Table 8.1.3.1.16.3.3-3  @sic R5-200150 sic@

    v_MeasObjectList := {cs_NR_MeasObjectToAddMod_NR(tsc_NR_MeasObjectId1, v_MeasObjectNR)};

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_RsrpThres, infinity_, -, -, cs_NR_MeasReportQuantity_Rsrp, -, true)}; // FFS MB reportInterval := ms480

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};

    v_MeasConfig :=cds_38508_MeasConfig_NR(v_MeasObjectList,

                                           v_ReportConfigList,

                                           v_MeasIdConfigList);

    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);

    //@siclog "Step 3" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in table 8.1.3.1.16.3.2-1/2.

    f_NR_SetCellPowerList(v_CellPowerLevels[1]);

    //@siclog "Step 4" siclog@

    //Check: does the UE transmit a MeasurementReport messages within the next 10s?

    f_NR_Check_NoMeasurementReport(10.0, nr_Cell1, cr_NR_MeasurementReport_Any);

    //@siclog "Steps 5-6" siclog@

    //The SS transmits an RRCReconfiguration message including both NR Cell 2 and NR Cell 4 in whiteCellsToAddModList

    //The UE transmits an RRCReconfigurationComplete message.

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour2 := f_NR_CellInfo_GetPhysicalCellId(nr_Cell2);

    v_PhysCellIdNeighbour4 := f_NR_CellInfo_GetPhysicalCellId(nr_Cell4);

    v_MeasObjectNR.whiteCellsToAddModList := {cs_PCI_RangeElement(v_PhysCellIdNeighbour2, tsc_PCI_RangeIndex1, n4),

                                              cs_PCI_RangeElement(v_PhysCellIdNeighbour4, tsc_PCI_RangeIndex2, n4)};

    v_MeasObjectList := {cs_NR_MeasObjectToAddMod_NR(tsc_NR_MeasObjectId1, v_MeasObjectNR)};

    v_MeasConfig := cds_38508_MeasConfig_NR(v_MeasObjectList,

                                            omit,

                                            omit);

    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);

    //@siclog "Step 7" siclog@

    //Wait and ignore MeasurementReport messages for 10s to allow UE measurement

    f_NR_IgnorePeriodicalMeasurmentReport(10.0);

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event A3 (measId 1) with the measured RSRP value for NR Cell 2?

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, 1, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour2, cr_NR_MeasQuantityResults_Rsrp)}));

    f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 8");

  }
After change

 function f_TC_8_1_3_1_16_TestBody() runs on NR_BASE_PTC

  {

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NR_CellPowerLevels_Type v_CellPowerLevels;

    var NR_AbsoluteCellPower_Type v_CellPower_A3_NoEntry_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_A3_Entry_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_ServingCell_FR1;

    var RSRP_Range v_RsrpThres;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour2;

    var PhysCellId v_PhysCellIdNeighbour4;

    var template (value) MeasObjectNR v_MeasObjectNR;

    var template (omit) MeasObjectToAddModList v_MeasObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    v_CellPower_ServingCell_FR1 := -85;

    v_CellPower_A3_NoEntry_FR1 := -91;

    v_CellPower_A3_Entry_FR1 := -78;

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_ServingCell_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),

                                cs_NR_CellPower(nr_Cell2, v_CellPower_A3_NoEntry_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),

                                cs_NR_CellPower(nr_Cell4, v_CellPower_A3_NoEntry_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell2, v_CellPower_A3_Entry_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    v_CellPowerLevels := {v_CellPowerList_AtT0, v_CellPowerList_AtT1};

    f_NR_SetCellPowerList(v_CellPowerLevels[0]);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup NR measurement and reporting for intra-frequency event A3 (measId 1)

    //The UE transmits an RRCReconfigrationComplete message

    v_RsrpThres := 0;

    v_MeasObjectNR := f_NR_MeasObjectNR_SSB (nr_Cell1, omit, omit);

    v_MeasObjectNR.whiteCellsToAddModList := {cs_PCI_RangeElement(900)};  //Table 8.1.3.1.16.3.3-3  @sic R5-200150 sic@

    v_MeasObjectList := {cs_NR_MeasObjectToAddMod_NR(tsc_NR_MeasObjectId1, v_MeasObjectNR)};

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_RsrpThres, infinity_, -, ms10240, cs_NR_MeasReportQuantity_Rsrp, -, true)}; // FFS MB reportInterval := ms480  //Script failure lc modify 20200321

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};

    v_MeasConfig :=cds_38508_MeasConfig_NR(v_MeasObjectList,

                                           v_ReportConfigList,

                                           v_MeasIdConfigList);

    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);

    //@siclog "Step 3" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in table 8.1.3.1.16.3.2-1/2.

    f_NR_SetCellPowerList(v_CellPowerLevels[1]);

    //@siclog "Step 4" siclog@

    //Check: does the UE transmit a MeasurementReport messages within the next 10s?

    f_NR_Check_NoMeasurementReport(30.0, nr_Cell1, cr_NR_MeasurementReport_Any);

    //@siclog "Steps 5-6" siclog@

    //The SS transmits an RRCReconfiguration message including both NR Cell 2 and NR Cell 4 in whiteCellsToAddModList

    //The UE transmits an RRCReconfigurationComplete message.

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour2 := f_NR_CellInfo_GetPhysicalCellId(nr_Cell2);

    v_PhysCellIdNeighbour4 := f_NR_CellInfo_GetPhysicalCellId(nr_Cell4);

    v_MeasObjectNR.whiteCellsToAddModList := {cs_PCI_RangeElement(v_PhysCellIdNeighbour2, tsc_PCI_RangeIndex1, n4),

                                              cs_PCI_RangeElement(v_PhysCellIdNeighbour4, tsc_PCI_RangeIndex2, n4)};

    v_MeasObjectList := {cs_NR_MeasObjectToAddMod_NR(tsc_NR_MeasObjectId1, v_MeasObjectNR)};

    v_MeasConfig := cds_38508_MeasConfig_NR(v_MeasObjectList,

                                            omit,

                                            omit);

    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);

    //@siclog "Step 7" siclog@

    //Wait and ignore MeasurementReport messages for 10s to allow UE measurement

    f_NR_IgnorePeriodicalMeasurmentReport(10.0);

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event A3 (measId 1) with the measured RSRP value for NR Cell 2?

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, 1, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour2, cr_NR_MeasQuantityResults_Rsrp)}));

    f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 8");

  }
3 Branches executed

4 Execution Log Files

4.1 HiSilicon Balong 5000
The HiSilicon Balong 5000 passed this test case on Datang LinkTester 5G Protocol Conformance Test System ECT9610. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file:
\ TC_8_1_3_1_16_NR5GC_N41.log
· PIXCIT settings used:
\ Pics(HiSilicon).xml

\Pixit(TC_8_1_3_1_16_NR5GC).xml
5 References

	[1]
	R5s200473:   Supporting information for agreement of NR5GC test case 8.1.3.1.16 in FR1


