
1. Overview

This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 7.1.2.3.11 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-06_D19wk38’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 5). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.2.3.11
Test Group:
\NR5GC\7_1_2\
ATS Version:
iwd-TTCN3-B2019-06_D19wk38
System Simulator used:
Anritsu Protocol Conformance Test System ME7834NR
UE used:
MediaTek MT6297
Verification Status:
PASS


4. Corrections required for NR5GC TC 7.1.2.3.11
Change 4.1 

	Function name
	f_TC_7_1_2_3_11_NR_TestBody

	Reason for change
	1. 38.508-1, Table 4.8.1-1A with condition RBConfig_NoKeyChange states that SRB RLC should be reestablished
2.  After Steps 1-4, SS only have ~310ms (t-pollretransmit or t-reassembly + sr-transmax) to do reconfiguration with sync procedure. Otherwise, UE will send RACH due to t-pollretransmit or t-reassembly expiry. Therefore, encoding of CellGroupConfig can be done before testcase starts to reduce unnecessary delays. RachProcedureConfig can also be done before start of testcase.
3. In f_NR_RLC_ReconfigurationHO, TTCN sets a contant delay timer tsc_NR_DelayBeforeIntraCellHO which is 300ms. For this timing sensitive testcase, it can be reduced to 200ms.

4. After reconfiguration with sync procedure, UE can transmit RRC Reconfiguration Complete and RLC SDU #1 for Step 8 at the same TTI. To ensure SS can deliver it to NR5GC_PTC without errors, altstep can be used. Step 7-8 is moved to testbody as TTCN will start default UL grant transmission after reconfiguration with sync procedure.

	Summary of change
	1. Add SRB1 and SRB2 in CellGroupConfig for reconfiguration with sync procedure in Step 5-6
2. Move encoding of CellGroupConfig and RachProcedureConfig before testcase starts. Use f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8 instead of f_NR_RLC_ReconfigurationHO
3. Set delay before intra-cell HO as 200ms
4. Use altstep for Steps 6 and 7

	TTCN module
	RLC_NR5GC.ttcn

	MCC160 Comment
	


Before change

  function f_TC_7_1_2_3_11_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                        DRB_Identity p_NR_DRB_Id) runs on NR5GC_PTC

  { // AM RLC / RLC re-establishment procedure

    //@sic R5-194296 sic@

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var NR_RLC_DataFieldList_Type v_NR_RLC_DataList;

    v_NR_RLC_DataList := f_TC_7_1_2_3_11_NR_Steps1To4(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Steps 5-6" siclog

    //The SS transmits NR RRCReconfiguration message to trigger RLC re-establishment on DRB using Reconfig with sync procedure

    //The UE transmits a NR RRCReconfigurationcomplete message.

    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_AM(p_NR_DRB_Id, -, ms150, -, -, -, -, ms150, -, true_);
    // Reconfiguration Msg to be transmitted on E-UTRA side for ENDC tc or NR side for NR5GC tc

    f_NR_RLC_ReconfigurationHO (nr_Cell1, p_NR_DRB_Id, v_RLC_BearerConfigList, cs_38508_PDCP_Config(-, -, -, ms160, omit));
    f_TC_7_1_2_3_11_NR_Steps7To15(p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_DataList);

  }  

After change
  function f_TC_7_1_2_3_11_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                        DRB_Identity p_NR_DRB_Id) runs on NR5GC_PTC

  { // AM RLC / RLC re-establishment procedure

    //@sic R5-194296 sic@

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var NR_RLC_DataFieldList_Type v_NR_RLC_DataList;

    var template (value) CellGroupConfig v_CellGroupConfig; 

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig; 

    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell1);

    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;

    var template (value) NR_RadioBearer_Type v_SS_Drb_Config;

    var template (value) DL_DCCH_Message v_RRCReconfiguration;

    var SubFrameTiming_Type v_Timing;

    var boolean v_RRC_MsgRcvd;

    var boolean v_UlData_Rcvd;

    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU_Expected;

    var template (present) NR_RLC_PDUList_Type v_NR_RLC_PDUList;

    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_AM(p_NR_DRB_Id, -, ms150, -, -, -, -, ms150, -, true_); 

    v_RLC_BearerConfigList[1] := cs_38508_RLC_BearerConfig_SRB(tsc_SRB1);

    v_RLC_BearerConfigList[1].reestablishRLC := true_;

    v_RLC_BearerConfigList[2] := cs_38508_RLC_BearerConfig_SRB(tsc_SRB2);

    v_RLC_BearerConfigList[2].reestablishRLC := true_;

    v_RRC_MsgRcvd:= false;

    v_UlData_Rcvd := false;

    v_CellGroupConfig := f_NR_GetCellGroupConfigHO(nr_Cell1, v_RLC_BearerConfigList); //tsc_C_RNTI_Value3

    v_RadioBearerConfig := f_NR5GC_GetExisting_RadioBearerConfig(false, cs_38508_PDCP_Config(-, -, -, ms160, omit)); // RBConfig_NoKeyChange

    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));

    v_RRCReconfiguration := cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext);

    v_SS_Drb_Config := cs_NR_SS_RadioBearer_AM_Test(p_NR_DRB_Id, -); // size18

    v_RachProcedureConfig := cs_NR_RachProcedureConfig_CRNTI_HO(v_NR_PhysicalParameters, tsc_C_RNTI_Value3);

    f_NR_SS_RachProcedureConfig(nr_Cell1, v_RachProcedureConfig);
    v_NR_RLC_DataList := f_TC_7_1_2_3_11_NR_Steps1To4(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Steps 5" siclog@

    //The SS transmits NR RRCReconfiguration message to trigger RLC re-establishment on DRB using Reconfig with sync procedure

    //The UE transmits a NR RRCReconfigurationcomplete message.

    // Reconfiguration Msg to be transmitted on E-UTRA side for ENDC tc or NR side for NR5GC tc

    v_Timing := f_NR_GetNextSendOccasion_DL(nr_Cell1, 200);

    f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(nr_Cell1, 

                                                  v_RRCReconfiguration, 

                                                  false, // RBConfig_NoKeyChange

                                                  {v_SS_Drb_Config}, 

                                                  v_RachProcedureConfig, 

                                                  cs_TimingInfo_NR(v_Timing), 

                                                  -, -, -, -, -, //tsc_C_RNTI_Value3

                                                  true);

    // @siclog "Steps 6-8" siclog@

    p_RLC_Rec.AM_RX_Next := 0;

    v_NR_RLC_AMD_PDU_Expected := f_NR_RLC_Get_AMD_FullSDU_RX(p_RLC_Rec, ?, v_NR_RLC_DataList[tsc_RLC_SDU1]);

    v_NR_RLC_PDUList := { cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_Expected) };

    alt {

        [not v_RRC_MsgRcvd] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))

        {

            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

            v_RRC_MsgRcvd := true;

            if(not v_UlData_Rcvd){ repeat;}

        }

        [not v_UlData_Rcvd] DRB.receive(car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1, p_NR_DRB_Id, -, v_NR_RLC_PDUList))

        {

            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7");

            v_UlData_Rcvd := true;

            if(not v_RRC_MsgRcvd){ repeat;}

        }

    };
    f_TC_7_1_2_3_11_NR_Steps7To15(p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_DataList, true);

  }
Change 4.2
	Function name
	f_TC_7_1_2_3_11_NR_Steps7To15

	Reason for change
	As stated in change 4.1.4, Step 7-8 is moved in testbody.

	Summary of change
	Skip Steps 7-8 in f_TC_7_1_2_3_11_NR_Steps7To15 as they are already done in f_TC_7_1_2_3_11_NR_TestBody

	TTCN module
	RLC_TC_Common_NR.ttcn

	MCC160 Comment
	


Before Changes 

    function f_TC_7_1_2_3_11_NR_Steps7To15(NR_RLC_SS_State_Type p_RLC_Rec,

                                         DRB_Identity p_NR_DRB_Id,

                                         NR_RLC_DataFieldList_Type p_NR_RLC_DataList

                                         ) runs on NR_BASE_PTC

  { //@sic R5-194296 sic@

    var integer v_RLC_Len1stSeg;

    var integer v_RLC_Len2ndSeg;

    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU;

    var template (present) NR_RLC_AM_StatusPDU_Type v_NR_RLC_StatusPDU;

    var template (present) NR_RLC_PDUList_Type v_NR_RLC_PDUList;

    var NR_RLC_DataField_Type v_RLC_Data1;

    timer t_WatchDog := 1.0;

    v_RLC_Len1stSeg := 20; // 160 bits

    v_RLC_Len2ndSeg := 20; // 160 bits

    // @siclog "Step 7" siclog@

    //The SS starts the UL default grant transmissions

    f_NR_ULGrantConfiguration_Start(nr_Cell1);

    // @siclog "Step 8" siclog@

    // The UE retransmits RLC SDU #1.

    p_RLC_Rec.AM_RX_Next := 0; // only update of the RLC record which is needed

    f_NR_RxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, p_NR_RLC_DataList[tsc_RLC_SDU1], ?);
    // @siclog "Step 9" siclog@

… 
}
After Changes 
   function f_TC_7_1_2_3_11_NR_Steps7To15(NR_RLC_SS_State_Type p_RLC_Rec,

                                         DRB_Identity p_NR_DRB_Id,

                                         NR_RLC_DataFieldList_Type p_NR_RLC_DataList,

                                         boolean p_Step7_8_DoneInTestbody := false
                                         ) runs on NR_BASE_PTC

  { //@sic R5-194296 sic@

    var integer v_RLC_Len1stSeg;

    var integer v_RLC_Len2ndSeg;

    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU;

    var template (present) NR_RLC_AM_StatusPDU_Type v_NR_RLC_StatusPDU;

    var template (present) NR_RLC_PDUList_Type v_NR_RLC_PDUList;

    var NR_RLC_DataField_Type v_RLC_Data1;

    timer t_WatchDog := 1.0;

    v_RLC_Len1stSeg := 20; // 160 bits

    v_RLC_Len2ndSeg := 20; // 160 bits

    if (not p_Step7_8_DoneInTestbody){
        // @siclog "Step 7" siclog@

        //The SS starts the UL default grant transmissions

        f_NR_ULGrantConfiguration_Start(nr_Cell1);

        // @siclog "Step 8" siclog@

        // The UE retransmits RLC SDU #1.

        p_RLC_Rec.AM_RX_Next := 0; // only update of the RLC record which is needed

        f_NR_RxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, p_NR_RLC_DataList[tsc_RLC_SDU1], ?);

    }
    // @siclog "Step 9" siclog@

…

  } 

Change 4.3
	Function name
	f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8

	Reason for change
	1. As stated in Change 4.1.2, RachProcedureConfig is moved to testbody before testcase starts to reduce unnecessary delays.
2. In Step 1 of f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8, PDCP count for SRBs and DRBs are stored before SS restablishes SRBs and DRBs. Step 7 will then restore PDCP count of SRBs (stored in Step 1). Because RRC Reconfiguration will be sent at Step 2, PDCP count for SRB1 DL should be incremented by 1.
3. Set timing info for f_NR_SS_PdcpCount_Set. Otherwise, PDCP count will not be properly restored.
4. Don’t refresh keys during activation of security as this testcase requires RBConfig_NoKeyChange

	Summary of change
	1. Move RachProcedureConfig before testcase starts
2. Increment PDCP count for SRB1 DL as RRC Reconfiguration will be sent

3. Set timing info for f_NR_SS_PdcpCount_Set
4. Add parameter in f_NR_SS_AS_ActivateSecurity_Current so that key refresh is optional

	TTCN module
	NR5GC_Handover.ttcn

	MCC160 Comment
	


Before Changes 

  function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_CellId,

                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,

                                                       boolean                                p_RBConfig_KeyChange := true,

                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,

                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,

                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,

                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                                       template (omit) MeasConfig             p_MeasConfig := omit,

                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,

                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,

                                                       boolean p_SetGetOfPDCPCount := true

                                                       ) runs on NR_BASE_PTC

  { //Added p_SetGetOfPDCPCount to be st to false in PDCP tc with TM mode

    var SubFrameTiming_Type v_Timing;

    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;

    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).

    // Not needd in PDCP tc with PDCP configured in TM mode

    if (p_SetGetOfPDCPCount){

        v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_CellId, p_NR_RadioBearerList, p_RBConfig_KeyChange);

    }

   /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.

    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */

    if (f_TimingInfo_IsNow(p_TimingInfo)) {

      v_Timing := f_NR_GetNextSendOccasion_DL(p_CellId, tsc_NR_DelayBeforeIntraCellHO);

    } else {

      v_Timing := valueof(p_TimingInfo.SubFrame);

    }

    //Step 2 At T: Send RRCReconfiguration.

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));

    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration

       => RLC ACK is not taken into consideration */

    //Step 3 At T + 5ms: Release SRBs and DRBs.

    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms

    f_NR_SS_SRBs_DRBs_Release(p_CellId, cs_TimingInfo_NR(v_Timing), v_DrbIdList);

    // Step 4 At T + 5ms: Configure RACH procedure either dedicated or C-RNTI based.

    // NOTE: Since the RACH procedure may require a new C-RNTI to be used it cannot be configured before sending out the RRCReconfiguration.

    f_NR_SS_C_RNTI_Config(p_CellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing)); //@sic R5s190747 change 5.3 sic@

    f_NR_CellInfo_SetRNTI(p_CellId, p_RNTI_Value);

    f_NR_SS_RachProcedureConfigHO(p_CellId, p_RachProcedureConfig, cs_TimingInfo_NR(v_Timing), p_RACH_ConfigDedicated);

    // Step 5 At T + 5ms: (Re-)configure measurement gap if provided in the RRCReconfiguration message.

    // @sic R5-197222 sic@

    if (isvalue(p_MeasConfig.measGapConfig)) {

        f_NR_SS_MeasGapCtrlConfig(p_CellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig), cs_TimingInfo_NR(v_Timing));

    }

    // Step 6 At T + 10ms: (Re-) configure SRBs and DRBs.

    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms

    f_NR_SS_SRBs_DRBs_Config(p_CellId, cs_TimingInfo_NR(v_Timing), p_NR_RadioBearerList);

    // Step 7 At T + 10ms:  Restore the PDCP count for AM DRBs and SRBs (if applied).

    //NOTE 2:   For AM DRBs the PDCP count is maintained while for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the  RRCReconfiguration message content

    // Not needed in PDCP tc with PDCP configured in TM mode

    if (p_SetGetOfPDCPCount){

        f_NR_SS_PdcpCount_Set(p_CellId, v_NR_PdcpCountInfoList);

    }

   // Step 8 At T + 10ms: Re-establish security, disable TA transmission.

   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.

   f_NR_SS_AS_ActivateSecurity_Current(p_CellId, p_NCC, cs_TimingInfo_NR(v_Timing));

   // Disable TA transmission

   f_NR_ULGrantConfiguration_Common(p_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);

  }
After Changes 
  function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_CellId,

                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,

                                                       boolean                                p_RBConfig_KeyChange := true,

                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,

                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,

                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,

                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                                       template (omit) MeasConfig             p_MeasConfig := omit,

                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,

                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,

                                                       boolean p_SetGetOfPDCPCount := true,

                                                       boolean p_SkipRachProcedureConfig := false
                                                       ) runs on NR_BASE_PTC

  { //Added p_SetGetOfPDCPCount to be st to false in PDCP tc with TM mode

    var SubFrameTiming_Type v_Timing;

    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;

    var integer v_PdcpCountSrb1Tx;
    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).

    // Not needd in PDCP tc with PDCP configured in TM mode

    if (p_SetGetOfPDCPCount){

        v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_CellId, p_NR_RadioBearerList, p_RBConfig_KeyChange);

    }

   /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.

    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */

    if (f_TimingInfo_IsNow(p_TimingInfo)) {

      v_Timing := f_NR_GetNextSendOccasion_DL(p_CellId, tsc_NR_DelayBeforeIntraCellHO);

    } else {

      v_Timing := valueof(p_TimingInfo.SubFrame);

    }

    //Step 2 At T: Send RRCReconfiguration.

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));

    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration

       => RLC ACK is not taken into consideration */

    //Step 3 At T + 5ms: Release SRBs and DRBs.

    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms

    f_NR_SS_SRBs_DRBs_Release(p_CellId, cs_TimingInfo_NR(v_Timing), v_DrbIdList);

    // Step 4 At T + 5ms: Configure RACH procedure either dedicated or C-RNTI based.

    // NOTE: Since the RACH procedure may require a new C-RNTI to be used it cannot be configured before sending out the RRCReconfiguration.

    f_NR_SS_C_RNTI_Config(p_CellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing)); //@sic R5s190747 change 5.3 sic@

    f_NR_CellInfo_SetRNTI(p_CellId, p_RNTI_Value);

    if (not p_SkipRachProcedureConfig){
        f_NR_SS_RachProcedureConfigHO(p_CellId, p_RachProcedureConfig, cs_TimingInfo_NR(v_Timing), p_RACH_ConfigDedicated);

    }
    // Step 5 At T + 5ms: (Re-)configure measurement gap if provided in the RRCReconfiguration message.

    // @sic R5-197222 sic@

    if (isvalue(p_MeasConfig.measGapConfig)) {

        f_NR_SS_MeasGapCtrlConfig(p_CellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig), cs_TimingInfo_NR(v_Timing));

    }

    // Step 6 At T + 10ms: (Re-) configure SRBs and DRBs.

    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms

    f_NR_SS_SRBs_DRBs_Config(p_CellId, cs_TimingInfo_NR(v_Timing), p_NR_RadioBearerList);

    // Step 7 At T + 10ms:  Restore the PDCP count for AM DRBs and SRBs (if applied).

    //NOTE 2:   For AM DRBs the PDCP count is maintained while for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the  RRCReconfiguration message content

    // Not needed in PDCP tc with PDCP configured in TM mode

    if (p_SetGetOfPDCPCount){

        v_PdcpCountSrb1Tx := bit2int(v_NR_PdcpCountInfoList[0].DL.Value) + 1;

        v_NR_PdcpCountInfoList[0].DL.Value := int2bit(v_PdcpCountSrb1Tx, 32);
        f_NR_SS_PdcpCount_Set(p_CellId, v_NR_PdcpCountInfoList, cs_TimingInfo_NR(v_Timing));

    }

   // Step 8 At T + 10ms: Re-establish security, disable TA transmission.

   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.

   f_NR_SS_AS_ActivateSecurity_Current(p_CellId, p_NCC, cs_TimingInfo_NR(v_Timing), p_RBConfig_KeyChange);

   // Disable TA transmission

   f_NR_ULGrantConfiguration_Common(p_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);

  }
Change 4.4
	Function name
	f_NR_SS_AS_ActivateSecurity_Current

	Reason for change
	As stated in Change 4.3.4, security keys should not be refreshed

	Summary of change
	Add option to not refresh security keys

	TTCN module
	NR_SecuritySteps.ttcn

	MCC160 Comment
	


Before Changes 

  function f_NR_SS_AS_ActivateSecurity_Current(NR_CellId_Type p_CellId,

                                               NextHopChainingCount p_NCC := tsc_NR_38508_NextHopChainingCount,

                                               template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now

                                               ) runs on NR_BASE_PTC

  {

    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();

    // refresh security keys

    v_RRC_SecurityParams := f_NR_InitAS_KeyChaining_KeyRefresh(f_NR_CellInfo_GetPhysicalCellId(p_CellId),

                                                               f_NR_CellInfo_GetFrequencySSB(p_CellId),

                                                               v_RRC_SecurityParams,

                                                               p_NCC);

    // reset the PDCP sequence numbers for ciphering

    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_CellId, false);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_RRC_SecurityParams.AS_Integrity,

                                          v_RRC_SecurityParams.AS_Ciphering,

                                          p_TimingInfo);

    f_NR_Security_Set(v_RRC_SecurityParams);

  }
After Changes 
  function f_NR_SS_AS_ActivateSecurity_Current(NR_CellId_Type p_CellId,

                                               NextHopChainingCount p_NCC := tsc_NR_38508_NextHopChainingCount,

                                               template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                               boolean p_RBConfig_KeyChange := true
                                               ) runs on NR_BASE_PTC

  {

    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();

    // refresh security keys

    if (p_RBConfig_KeyChange){
        v_RRC_SecurityParams := f_NR_InitAS_KeyChaining_KeyRefresh(f_NR_CellInfo_GetPhysicalCellId(p_CellId),

                                                                    f_NR_CellInfo_GetFrequencySSB(p_CellId),

                                                                    v_RRC_SecurityParams,

                                                                    p_NCC);

    }
    // reset the PDCP sequence numbers for ciphering

    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_CellId, false);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_RRC_SecurityParams.AS_Integrity,

                                          v_RRC_SecurityParams.AS_Ciphering,

                                          p_TimingInfo);

    f_NR_Security_Set(v_RRC_SecurityParams);

  }
Change 4.5
	Function name
	cs_NR_RachProcedureConfig_CRNTI_HO

	Reason for change
	Wrong uplink grant size for DCI format 0_1_And_1_1

	Summary of change
	Set correct UL grant size

	TTCN module
	NR5GC_Handover.ttcn

	MCC160 Comment
	


Before Changes 

    template (value) NR_RachProcedureConfig_Type cs_NR_RachProcedureConfig_CRNTI_HO(NR_PhysicalParameters_Type p_PhysicalParameters,

                                                                                  RNTI_Value_Type p_TempC_RNTI_Value := tsc_C_RNTI_Value3) :=

  { /* Acc to 38.523-3 cl 7.3.5.3.3  */

    RachProcedureList := {{

        RandomAccessResponse := cs_NR_RandomAccessResponseConfig(p_PhysicalParameters, cs_NR_RAR_SubPdu_RapIdAndPayload(-,

                                                                                                                        tsc_NR_RACH_TA_Def,

                                                                                                                        f_NR_RAR_UplinkGrant(p_PhysicalParameters.UL_BWPs[0], 48), // Table 7.1.2.3.3-1: Default grants

                                                                                                                        cs_NR_TempC_RNTI_Explicit(p_TempC_RNTI_Value))),

        ContentionResolution := cs_NR_ContentionResolutionCtrl_CrntiBased(p_PhysicalParameters, f_NR_ResourceAllocation_UplinkGrant(128, p_PhysicalParameters.UL_BWPs[0])) // Table 7.1.2.3.3-1: Default grants

    }}

  };
After Changes 
    template (value) NR_RachProcedureConfig_Type cs_NR_RachProcedureConfig_CRNTI_HO(NR_PhysicalParameters_Type p_PhysicalParameters,

                                                                                  RNTI_Value_Type p_TempC_RNTI_Value := tsc_C_RNTI_Value3) :=

  { /* Acc to 38.523-3 cl 7.3.5.3.3  */

    RachProcedureList := {{

        RandomAccessResponse := cs_NR_RandomAccessResponseConfig(p_PhysicalParameters, cs_NR_RAR_SubPdu_RapIdAndPayload(-,

                                                                                                                        tsc_NR_RACH_TA_Def,

                                                                                                                        f_NR_RAR_UplinkGrant(p_PhysicalParameters.UL_BWPs[0], 48), // Table 7.1.2.3.3-1: Default grants

                                                                                                                        cs_NR_TempC_RNTI_Explicit(p_TempC_RNTI_Value))),

        ContentionResolution := cs_NR_ContentionResolutionCtrl_CrntiBased(p_PhysicalParameters, f_NR_ResourceAllocation_UplinkGrant(136, p_PhysicalParameters.UL_BWPs[0])) // Table 7.1.2.3.3-1: Default grants

    }}

  };
5. Execution Log Files

5.1 MediaTek MT6297
The MediaTek MT6297 UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR on NR5GC band combination N78.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

